
【概要】 ICOMのCI-Vコマンドを使い、FT8で運用する周波数を巡回するメモリスキャン機能を 
Excel VBA（Visual Basic for Applications）でプログラミングしたものを紹介する。 

【構成】 本Excel VBA 
VSPE(仮想シリアル共有） 

FT8 Soft 
Rig（ICOM製） 

COM1 9600bps USB 
COM n 

Memory Scan by Serial CTL (COM1 9600) for ICOM

Fn + DEL (Break)で停止

Cycle 0:01 :00 h:mm:ss

Ch No Freq (MHz) Scan
1 1.840
2 3.531
3 3.573
4 7.041
5 7.074
6 10.136
7 14.074
8 18.100
9 21.074

10 24.915
11 28.074
12 50.313
13 50.323
14 144.460
15 430.510
16 1296.600

開始

入力画面 

スキャン時間の設定可 

周波数は任意に設定可 
 
チェックBOXで 
スキャン有無を設定可 

動作フロー 

入力画面の情報取得 

開始時刻調整（FT8用に毎15秒単位） 

Ch No セット(i=1～16)の繰り返し 

周波数情報をCI-Vフォーマットに変換 

スキャン有無を
チェック 

COMポートから送信 

FE FE DST EO 00 f1 f2 f3 f4 f5 FD 

CI-Vフォーマット 

スキャン時間待ち 

機種コード 周蓮設定 

周蓮データ 

(Virtual Serial Port Emulator) 



クラスモジュール(clsCOM_Port)のコード（windowsのファンクションコール）は以下のURLにあるコードをコピーして使用 
https://qiita.com/takeru0x5569/items/21c7fcf815de6a96a5b3 Sub Main() 

 
'### 変数の型宣言 ### 
 
Dim Wait_time As Date 
Dim Memory_Freq() As Variant 
Dim Flag() As Variant 
 
Dim newHour As Integer 
Dim newMinute As Integer 
Dim newSecond As Integer 
Dim Start_time As Date 
 
Dim i As Integer 
Const MAX_Ch As Integer = 16 
 
Dim Memory_Data As String 
Dim Freq(4) As Byte 
 
'### セル値の取得 ### 
 
Wait_time = Range("C5").Value 
Memory_Freq = Range("C8:C23").Value 
Flag = Range("E8:E23").Value 
 
'### 開始時刻調整 ### 
 
newHour = Hour(Now()) 
newMinute = Minute(Now()) 
newSecond = (Second(Now()) ¥ 15 + 1) * 15 + 1 
 
Start_time = TimeSerial(newHour, newMinute, newSecond) 
Application.Wait Start_time 
 
'### スキャン開始 ### 
 
Do While (1) '←無限ループ 
    For i = 1 To MAX_Ch 
      Application.StatusBar = "Scaning Ch " & i 
        If Flag(i, 1) Then 
           Memory_Data = Format(Memory_Freq(i, 1), "0000.000000") 
             Freq(0) = "&H" & (Mid(Memory_Data, 10, 2)) 
             Freq(1) = "&H" & (Mid(Memory_Data, 8, 2)) 
             Freq(2) = "&H" & (Mid(Memory_Data, 6, 2)) 
             Freq(3) = "&H" & (Mid(Memory_Data, 3, 2)) 
             Freq(4) = "&H" & (Mid(Memory_Data, 1, 2)) 
               Call testCOM(Freq) 
             Application.Wait Now() + TimeValue(Wait_time) 
        End If 
    Next i 
Loop 
 
End Sub 

Sub testCOM(ByRef Freq() As Byte) 
 
    '### クラス宣言とインスタンス化 ### 
    Dim CP As clsCOM_Port 
    Set CP = New clsCOM_Port 
     
    Dim ByteBuffer(10) As Byte 
     
    '### COMポートを開く ### 
    Call CP.OpenCOM("COM1", 9600) '引数:COM番号,ボーレート 
     
    '### バッファクリア ### 
    Call CP.ClearBuffer 
         
    '### 送信文字列設定 ### 
    ByteBuffer(0) = &HFE 'プリアンブル 
    ByteBuffer(1) = &HFE 'プリアンブル 
   'ByteBuffer(2) = &H8E '送信先 (IC-7851:8E) 
   'ByteBuffer(2) = &H98 '送信先 (IC-7610:98) 
   'ByteBuffer(2) = &H94 '送信先 (IC-7300:94) 
   'ByteBuffer(2) = &HA4 '送信先 (IC-705 :A4) 
    ByteBuffer(2) = &HA2 '送信先 (IC-9700:A2) 
    ByteBuffer(3) = &HE0 '送信元（PC) 
    ByteBuffer(4) = &H0  'コマンド 周波数データの設定 
     
    '### 設定周波数(BCD 5byte) ### 
    ByteBuffer(5) = Freq(0) 
    ByteBuffer(6) = Freq(1) 
    ByteBuffer(7) = Freq(2) 
    ByteBuffer(8) = Freq(3) 
    ByteBuffer(9) = Freq(4) 
     
    ByteBuffer(10) = &HFD 'ポストアンブル 
     
    '### 文字列送信 ### 
    Call CP.SendByte(ByteBuffer) 
 
    '### COMポートを閉じる ### 
    CP.CloseCOM 
         
End Sub 

VBA コード（標準モジュール） 


